# $Id: PKGBUILD 168990 2012-10-17 10:19:13Z allan $
# Maintainer :
# Contributor: Hugo Doria <hugo@archlinux.org>

pkgname=bzr
pkgver=2.5.1
pkgrel=2
pkgdesc="A decentralized revision control system (bazaar)"
arch=('i686' 'x86_64')
url="http://bazaar.canonical.com/en/"
license=('GPL')
depends=('python2')
optdepends=('python2-paramiko: for sftp support')
source=("http://launchpad.net/${pkgname}/${pkgver%.*}/${pkgver}/+download/${pkgname}-${pkgver}.tar.gz"{,.sig})
md5sums=('ac5079858364a046071000d5cdccb67b'
         'c83dd08dd2c3e27edbab873500d7d6eb')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  sed -i 's|man/man1|share/man/man1|' setup.py
  sed -i 's|/usr/bin/env python|/usr/bin/env python2|' \
    bzrlib/{plugins/bash_completion/bashcomp.py,tests/ssl_certs/create_ssls.py,patiencediff.py,_patiencediff_py.py}
  python2 setup.py build
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  python2 setup.py install --prefix=/usr --root="${pkgdir}" --optimize=1

  # bash-completion
  install -D -m644 contrib/bash/bzr \
    "${pkgdir}/usr/share/bash-completion/completions/bzr"
}
